

COMA

Regelungstechnik mit Maxima

(Control engineering with MAXima)

Wilhelm Haager
HTL St. Pölten, Abteilung Elektrotechnik
wilhelm.haager@htlstp.ac.at

Version 1.8, 18. April 2017

Inhaltsverzeichnis

1	Einführung	3
1.1	Begriffe	3
1.2	wxMaxima Benutzeroberfläche	4
1.3	Grundlegendes zum Paket COMA	4
2	Plotroutinen	5
2.1	Optionen des Gnuplot-Interfaces Draw	5
2.2	Zusätzliche Optionen von COMA	6
2.3	Plot	6
2.4	Isolinien	8
3	Übertragungsfunktionen	9
4	Laplace Transformation, Sprungantwort	13
5	Frequenzgänge	15
6	Untersuchungen in der s-Ebene	20
6.1	Pol/Nullstellen-Verteilung	20
6.2	Wurzelortskurven	21
7	Stabilitätsverhalten	23
7.1	Stabilität	23
7.2	Stabilitätsgüte	25
8	Optimierung	27
9	Reglerentwurf	28
10	Zustandsraum	30
11	Diverse Funktionen	34
	Literaturverzeichnis	35

1 Einführung

1.1 Begriffe

Maxima: Open-Source Abkömmling des Computeralgebra-Systems *Macsyma*, das ursprünglich 1967–1982 im Auftrag des US-Energieministeriums am MIT entwickelt wurde. 1989 wurde eine Version von *Macsyma* mit dem Namen *Maxima* unter der *GNU General Public Licence* veröffentlicht und wird nun von einer unabhängigen Gruppe von Anwendern weiterentwickelt. *Maxima* ist in Lisp geschrieben und enthält selbst viele Elemente der funktionalen Programmierung.

Aufgrund seiner Leistungsfähigkeit und freien Verfügbarkeit gibt es eigentlich keinen Grund, es *nicht* zu verwenden.

wxMaxima: Eine von mehreren grafischen Benutzeroberflächen für *Maxima*. Es ermöglicht die Eingabe und das Editieren von Ausdrücken, sowie das Dokumentieren von Rechengängen mit Text und Bildern in einem Arbeitsfenster. Rechenergebnisse und (auf Wunsch) Grafiken werden von *Maxima* in dieses Arbeitsfenster ausgegeben.

Arbeitssitzungen können abgespeichert, geladen und wiederausgeführt werden; die wichtigsten Befehle können (für notorische Mausklickser) über Menüs und Befehlsschaltflächen aufgerufen werden. Arbeitssitzungen können als eine HTML-Filestruktur oder als \LaTeX -File exportiert werden. Beim Export nach \LaTeX ist mitunter etwas händisches Nacheditieren des resultierenden TEX-Files notwendig.


Bei der Installation von *Maxima* wird *wxMaxima* automatisch als Standard-Interface mitinstalliert.

COMA (*CO*ntrol *en*gineering with *MA*xima):

Regelungstechnik-Paket für *Maxima*, enthält grundlegende Verfahren zur Systemanalyse im Zeit-, Frequenz- und Laplacebereich, Reglerentwurfsverfahren, sowie Verfahren im Zustandsraum:

- Laplace-Rücktransformation von Übertragungsfunktionen beliebiger Ordnung (die in *Maxima* eingebaute Funktion versagt im Allgemeinen bei höherer als zweiter Ordnung)
- Sprungantworten
- Ortskurven und Bodediagramme
- Pol- und Nullstellen, Wurzelortskurven
- Stabilitätsuntersuchungen: Stabilitätsgrenze, Hurwitzkriterium, Stabilitätsbereiche in der Parameterebene, Phasenrand, Amplitudenrand
- Optimierung und Reglerentwurf: ISE-Kriterium, Betragsoptimum
- Zustandsraum: Umwandlung in eine Übertragungsfunktion, Normalformen, Steuerbarkeit, Beobachtbarkeit

1.2 wxMaxima Benutzeroberfläche



- ① ... Arbeitsbereich zur Ein- und Ausgabe
 ② ... Gnuplot-Ausgabefenster

1.3 Grundlegendes zum Paket COMA

- Die Laplacevariable ist immer s , die Zeitvariable immer t ; bei Funktionen die den Frequenzgang betreffen, ist die Kreisfrequenz ω . Bei Frequenzgängen wird s gegebenenfalls automatisch durch $j\omega$ ersetzt. Übertragungsfunktionen sind gebrochen rationale Funktionen in s , Totzeiten werden nicht unterstützt, können aber mit Padé-Näherungen approximiert werden.
- Allen Funktionen, die eine Übertragungsfunktion als Parameter übernehmen, kann auch (ohne besondere Erwähnung) eine Liste von Übertragungsfunktionen als Parameter übergeben werden; der Rückgabewert ist dann ebenfalls eine Liste, deren einzelne Elemente den jeweiligen Übertragungsfunktionen entsprechen. Dies ist insbesondere bei Grafiken wichtig, wenn mehrere Kurven in ein einziges Diagramm gezeichnet werden sollen.
 Die zu plottende Liste kann nicht nur Funktionen (Übertragungsfunktionen) enthalten, sondern auch Grafikobjekte des Gnuplot-Interfaces *Draw* (*explicit*, *points*, *implicit*, *parametric*, *polar*, *polygon*, *rectangle*, *ellipse*, *label*). Damit können beispielsweise Diagramme mit Beschriftungen und sonstigen Grafikelementen versehen werden, sowie berechnete Kurvenverläufe mit Messpunkten verglichen werden.
- Zusätzlich zu den zu plottenden Funktionen können alle Plotroutinen optionale Parameter der Form *option = value* enthalten, mit denen die Grafik in Bezug auf Strichstärken, Größe, Darstellungsbereich, Ausgabebereich etc. konfiguriert werden kann.
- Vor der erstmaligen Verwendung ist das Paket mit dem Befehl `load` zu laden.

2 Plotroutinen

Zu Darstellung von Grafiken verwendet Maxima das Programm *Gnuplot* [2], das bei der Erstellung der Grafik aufgerufen wird. Die Grafik wird dabei entweder in einem eigenen Gnuplot-Fenster dargestellt (mit der Option `wx=false`), oder direkt im Arbeitsfenster von `wxMaxima` (mit der Option `wx=true`, default).

Die Plotroutinen von *COMA* verwenden nicht die Standardfunktionen von Maxima (`plot2d`, `plot3d`, `wxplot2d`, `wxplot3d`), sondern die Funktionen des Zusatzpaketes *Draw* (`draw2d`, `draw3d`, `wxdraw2d` und `wxdraw3d`), siehe [1], [3]. Diese sind zwar etwas komplizierter in der Anwendung, bieten aber mehr Möglichkeiten, die Grafiken mit Hilfe von Optionen an die eigenen Wünsche anzupassen.

Alle Plotroutinen übernehmen als Parameter eine einzige Funktion oder eine *Liste* von Funktionen; zusätzliche optionale Parameter in der Form `option=value`. Optionen, die einzelne Grafikobjekte betreffen (`color`, `line_width` etc.), können auch in einer Liste angegeben werden, deren Elemente den einzelnen Grafikobjekten zugeordnet werden: `option=[val1, val2, ...]`.

2.1 Optionen des Gnuplot-Interfaces Draw

<code>terminal=target</code>	Ziel für die Grafikausgabe, mögliche Werte: <code>screen</code> (default), <code>jpg</code> , <code>png</code> , <code>pngcairo</code> , <code>eps</code> , <code>eps_color</code>
<code>file_name=string</code>	Name des Ausgabefiles, Default: <code>maxima_out.ext</code>
<code>color=c</code>	Linienfarbe
<code>line_width=w</code>	Strichstärke
<code>xrange=[x₁, x₂]</code>	Darstellungsbereich in x-Richtung
<code>yrange=[y₁, y₂]</code>	Darstellungsbereich in y-Richtung
<code>zrange=[z₁, z₂]</code>	Darstellungsbereich in z-Richtung
<code>logx=true/false</code>	logarithmische Skalierung der x-Achse
<code>logy=true/false</code>	logarithmische Skalierung der y-Achse
<code>logz=true/false</code>	logarithmische Skalierung der z-Achse
<code>grid=true/false</code>	Darstellung eines Koordinatengitters
<code>enhanced3d=true/false</code>	Einfärbung der Fläche bei 3D-Plots

Wichtige Optionen des Gnuplot-Interfaces Draw

Eine vollständige Liste der Optionen findet sich im Maxima-Manual [1].

2.2 Zusätzliche Optionen von COMA

<code>wx=true/false</code>	legt die Art der Ausgabe fest: true ... Ausgabe in das wxMaxima-Arbeitsfenster false ... Ausgabe in ein Gnuplot-Fenster
<code>aspect_ratio=value</code>	Verhältnis Höhe/Breite des Diagramms; der Wert -1 bewirkt gleiche Skalierung von x-Achse und y-Achse
<code>color=[c₁,c₂,...]</code>	Liste von Farbwerten, die in entsprechender Reihenfolge für die einzelnen zu plottenden Elemente angewandt werden
<code>line_width=[w₁,w₂,...]</code>	Liste von Strichstärken für die einzelnen Elemente
<code>dimensions=[width,height]</code>	Breite und Höhe der Grafik
	⋮
<i>Globale Variable:</i>	
<code>coma_defaults</code>	Liste mit Optionen der Form <i>option=value</i>

Zusätzliche Optionen von COMA

Die Variable `coma_defaults` ist eine Liste mit Defaultwerten für Plot-Einstellungen in Form von Schlüssel-Wert Paaren. Im Gegensatz zum Befehl `set_draw_defaults` des Gnuplot-Interfaces *Draw* kann `coma_defaults` auch andere Optionen enthalten, die *nicht* Bestandteil von *Draw* sind.

2.3 Plot

Die Funktion `plot` dient zur xy-Darstellung von Funktionen $f(x)$ in einer Variablen oder zur dreidimensionalen Darstellung von Funktionen $f(x,y)$ in zwei Variablen.

<code>plot(f(x), opts)</code>	Plotten der Funktion $f(x)$ in einem zweidimensionalen Koordinatensystem
<code>plot(f(x,y), opts)</code>	Plotten der Funktion $f(x,y)$ in 3D-Darstellung
	Anstelle einer einzigen Funktion f kann auch eine <i>Liste</i> von Funktionen $[f_1, f_2, \dots]$ geplottet werden.

Plotfunktion für 2D-und 3D-Grafiken

Intern werden die Funktionen des Paketes *Draw* (`wxdraw2d` oder `draw2d` bzw. `wxdraw3d` oder `draw3d`) mit entsprechenden Parametern aufgerufen. Der (bequeme) Aufruf von

```
plot([f(x),g(x)], xrange=[0,10], color=[red,blue])
```

entspricht genau dem (weniger bequem aufzurufenden) Befehl

```
wxdraw2d(xrange=[0,10], color=red, explicit(f(x),x,0,10),
color=blue, explicit(g(x),x,0,10)).
```

Der Wert der Option `aspect_ratio`, die es in dieser Form in den Routinen des Paketes *Draw* nicht gibt, wird in der Option `user_preamble` in entsprechender Form an Gnuplot weitergegeben.

Laden des Regelungstechnik-Pakets

```
(%i1) load(coma)$
coma v.1.84, (Wilhelm Haager, 2017-02-24)
```

Defaultwerte für die Grafik-Optionen


```
(%i2) coma_defaults ;
(%o2) [grid=true,wx=true,dimensions=[600,350],
user_preamble=set grid linetype 3 lc '#444444',
color=[red,blue,dark-green,goldenrod,violet,
gray40,dark-cyan,orange,brown,sea-green]]
```

Liste mit Farbwerten

```
(%i3) col:[red,green,brown,blue];
(%o3) [red,green,brown,blue]
```


Plotten einer Liste von vier Funktionen in jeweils *einer einzigen* Variablen; intern wird die Funktion `wxdraw2d` aufgerufen. Die Variablennamen können in jeder Funktion auch unterschiedlich sein.

```
(%i4) plot([sin(5*x)**2,0.8*sin(5*y)**2,x,0.8*y],
xrange=[-0.5,1.5],color=col,
line_type=[solid,solid,dots,dots]);
```



Plotten einer Liste von zwei Funktionen in jeweils *zwei* Variablen; intern wird die Funktion `wxdraw3d` aufgerufen. Die Option `surface_hide=true` unterdrückt verdeckte Linien.

```
(%i5) plot([sin(5*x)**2+0.8*sin(5*y)**2,x+0.8*y],
xrange=[-0.5,1.5],surface_hide=true)$
```



`plot` evaluiert die zu plottende Funktion f vor der Berechnung einzelner Kurvenpunkte. Soll die Funktion für jeden einzelnen Kurvenpunkt neu evaluiert werden, ist sie zu *quotieren*. Das ist z.B. für Kenngrößen von Übertragungsfunktionen notwendig (siehe Abschnitt 7), die nur numerisch berechenbar sind.

Übertragungsfunktion mit einer Abhängigkeit von einem Parameter a


```
(%i6) f:(s+a)/(s^3+a*s^2+2*s+a);
(%o6) 
$$\frac{s+a}{s^3+a s^2+2 s+a}$$

```

Die Dämpfung kann nur *numerisch* berechnet werden, was nur dann gelingt, wenn a einen fixen Wert hat. f muss daher quotiert werden, um eine frühzeitige Evaluation zu verhindern.

```
(%i7) plot('damping_ratio(f),xrange=[0,5]);
```

```
(%t7)
```



2.4 Isolinien

Die Funktion `contourplot` zeichnet Isolinien einer Funktion $f(x,y)$. Im Gegensatz zur Funktion `contour_plot`, die Bestandteil von Maxima ist, verwendet sie wie alle Plotroutinen des Paketes COMA das Gnuplot-Interface *Draw* und hat auch die selben Optionen.

`contourplot(f(x,y), x, y, opts)` Plotten von Isolinien der Funktion $f(x,y)$
`contours=[z1, z2, ...]` Festlegen der Funktionswerte für die Isolinien

Isolinien

Übertragungsfunktion mit zwei freien Parametern a und b

```
(%i8) f:1/(s^5+s^4+5*s^3+a*s^2+b*s+1);
```

```
(%o7)
```


```
(%o8) 
$$\frac{1}{s^5 + s^4 + 5s^3 + as^2 + bs + 1}$$

```

Isolinien für die Dämpfung in Abhängigkeit der Parameter a und b . Die schwarze Linie bei der Dämpfung 0 kennzeichnet die Stabilitätsgrenze, die grünen Linien kennzeichnen den stabilen Bereich, die roten den instabilen Bereich.

```
(%i9) contourplot('damping(f),a,b,xrange=[0,8],  
yrange=[0,7],contours=[-0.3,-0.2,-0.1,0,0.1],  
color=[red,red,red,black,green],  
ip_grid_in=[20,20])$
```

```
(%t9)
```



3 Übertragungsfunktionen

Zur bequemen Erzeugung von Übertragungsfunktionen, in erster Linie zu Testzwecken und zum Experimentieren, stellt COMA folgende Funktionen bereit:

<code>rantranf(n)</code>	Zufalls-Übertragungsfunktion n -ter Ordnung, deren Zähler- und Nennerkoeffizienten Zufallswerte zwischen 1 und 10 sind
<code>stable_rantranf(n)</code>	Stabile Zufalls-Übertragungsfunktion (nur bis 6. Ordnung)
<code>gentranf(c, k, d, n)</code>	Übertragungsfunktion n -ter Ordnung, (Zählerpolynom k -ter Ordnung) mit den Zählerkoeffizienten c_i und den Nennerkoeffizienten d_i
<code>tranftype(F(s))</code>	Typ der Übertragungsfunktion $F(s)$ als Zeichenkette
<code>ntranfp(F(s))</code>	Liefert <code>true</code> , wenn alle Koeffizienten der Übertragungsfunktion $F(s)$ Zahlen (und keine symbolischen Ausdrücke) sind.
<code>closed_loop(Fo(s))</code>	Übertragungsfunktion $F_W(s)$ des geschlossenen Regelkreises aus der offenen Regelschleife $F_O(s)$
<code>open_loop(Fw(s))</code>	Übertragungsfunktion $F_O(s)$ der offenen Regelschleife aus dem geschlossenen Regelkreis $F_W(s)$
<code>time_delay(T, n, [k])</code>	Padé-Näherung für ein Totzeitsystem der Ordnung n . Die Zählerordnung k ist optional.
<code>impedance_chain(Z₁, Z₂, ... [n])</code>	Übertragungsfunktion einer Impedanzkette mit den Impedanzen Z_1, Z_2, \dots und einem (optionalen) Wiederholungsfaktor n
<code>transfer_function(eqs, vars, u, y)</code>	Berechnung der Übertragungsfunktion aus den Gleichungen eqs in den Variablen $vars$ mit der Eingangsgröße u und der Ausgangsgröße y
<code>sum_form(F(s), [n])</code>	Summendarstellung von $F(s)$ (eine von vier Möglichkeiten in Abhängigkeit von n)
<code>product_form(F(s), [n])</code>	Produktdarstellung von $F(s)$ (Aufspalten in lineare und quadratische Faktoren)

Erzeugung von Übertragungsfunktionen

Die Funktion `stable_rantranf(n)` sucht per Zufallsgenerator so lange Nennerkoeffizienten zwischen 1 und 10, bis eine *stabile* Übertragungsfunktion gefunden wird, was mit steigender

Systemordnung immer schwieriger wird, die Rechenzeit steigt bei siebter Ordnung extrem an. `stable_rantrnf` funktioniert daher nur bis zur 6. Ordnung.

Höhere Ordnungen können durch Multiplikation mehrerer Übertragungsfunktionen erreicht werden, die Koeffizienten sind dann allerdings nicht mehr auf den Bereich 1...10 beschränkt.

Erzeugung einer Liste von Zufalls-Übertragungsfunktionen vierter Ordnung, die Zählerordnung ist mindestens um eins niedriger.

```
(%i1)  fli:makelist(rantrnf(3),k,1,4);
(%o1)  [  $\frac{4s^2+8s+10}{4s^3+s^2+4s+10}$ ,  $\frac{9s+10}{7s^3+7s^2+2s+4}$ ,
 $\frac{2s^2+10s+7}{4s^3+5s^2+9s+7}$ ,  $\frac{2s^2+s+10}{10s^3+10s^2+2s+5}$  ]
```

Prüfung der Stabilität der Übertragungsfunktionen (siehe Abschnitt 7)

```
(%i2)  stablep(fli);
(%o2)  [false, false, true, false]
```

Erzeugung einer Liste von stabilen Zufalls-Übertragungsfunktionen

```
(%i3)  fli:makelist(stable_rantrnf(3),k,1,4);
(%o3)  [  $\frac{3}{2s^3+10s^2+7s+4}$ ,  $\frac{4s^2+10s+1}{2s^3+3s^2+6s+4}$ ,
 $\frac{8s+7}{2s^3+9s^2+s+1}$ ,  $\frac{10}{4s^3+10s^2+4s+5}$  ]
```

Alle sind stabil.

```
(%i4)  stablep(fli);
(%o4)  [true, true, true, true]
```

Erzeugung einer Liste von Übertragungsfunktionen

```
(%i5)  fo:[k/s, 5/(s*(s+3)), 1-b/s];
(%o5)  [  $\frac{k}{s}$ ,  $\frac{5}{s(s+3)}$ ,  $1-\frac{b}{s}$  ]
```

Berechnung der Übertragungsfunktionen der geschlossenen Regelkreise

```
(%i6)  fw:closed_loop(fo);
(%o6)  [  $\frac{k}{s+k}$ ,  $\frac{5}{s^2+3s+5}$ ,  $\frac{s-b}{2s-b}$  ]
```

Ermittlung des Typs der Übertragungsfunktionen als Textstrings

```
(%i7)  tranftype(fw);
(%o7)  [PT1, PT2, PDT1]
```

Überprüfung, ob alle Koeffizienten der Übertragungsfunktionen numerische Werte sind

```
(%i8)  ntranfp(fw);
(%o8)  [false, true, false]
```

Zurückrechnung auf die offenen Regelschleifen

```
(%i9)  open_loop(fw);
(%o9)  [  $\frac{k}{s}$ ,  $\frac{5}{s^2+3s}$ ,  $\frac{s-b}{s}$  ]
```

`gentranf(a,k,b,n)` erzeugt eine allgemeine Übertragungsfunktion mit indizierten Koeffizienten der Form

$$\frac{a_0 + a_1s + a_2s^2 + \dots + a_k s^k}{b_0 + b_1s + b_2s^2 + \dots + b_n s^n}.$$

Übertragungsfunktion mit allgemeinen Koeffizienten a_i und b_i

```
(%i10) gentranf(a,3,b,5);
(%o10) 
$$\frac{a_3 s^3 + a_2 s^2 + a_1 s + a_0}{b_5 s^5 + b_4 s^4 + b_3 s^3 + b_2 s^2 + b_1 s + b_0}$$

```


Totzeitsysteme haben transzendente Übertragungsfunktionen, eine Rücktransformation totzeit-behafteter Regelkreise ist im Allgemeinen in geschlossener Form nicht möglich.

`time_delay(T,n,k)` liefert eine Padénäherung n -ter Ordnung für ein Totzeitsystem mit der Übertragungsfunktion $G(s) = e^{-sT}$. Die Angabe k der Ordnung des Zählerpolynoms ist optional, ihr Defaultwert ist $n - 1$.

Padénäherung vierter Ordnung für die Übertragungsfunktion einer Totzeit $G(s) = \exp(-sT)$

$$\begin{aligned} (\%i11) \quad & \text{time_delay}(T,4); \\ (\%o11) \quad & \frac{-4 T^3 s^3 + 60 T^2 s^2 - 360 T s + 840}{T^4 s^4 + 16 T^3 s^3 + 120 T^2 s^2 + 480 T s + 840} \end{aligned}$$

`impedance_chain` erzeugt die Übertragungsfunktion einer Impedanzkette mit beliebig vielen Impedanzen (gerader Anzahl); der letzte (optionale ganzzahlige) Parameter gibt an, wie oft sich die Impedanzkette wiederholt.



Übertragungsfunktion einer Impedanzkette mit vier Elementen


$$\begin{aligned} (\%i12) \quad & \text{impedance_chain}(R,1/(s*C),s*L+R,1/(s*C)); \\ (\%o12) \quad & \frac{1}{C^2 L R s^3 + (C^2 R^2 + C L) s^2 + 3 C R s + 1} \end{aligned}$$

Übertragungsfunktion einer Impedanzkette mit viermal den gleichen Elementen

$$\begin{aligned} (\%i13) \quad & \text{impedance_chain}(R,1/(s*C),4); \\ (\%o13) \quad & \frac{1}{C^4 R^4 s^4 + 7 C^3 R^3 s^3 + 15 C^2 R^2 s^2 + 10 C R s + 1} \end{aligned}$$

`transfer_function(eqs,vars,u,y)` berechnet die Übertragungsfunktion aus einer Liste mit linearen Gleichungen `eqs`, beispielsweise aus einem Blockschaltbild. `vars` ist die Liste der verwendeten Variablen, nach denen das Gleichungssystem aufzulösen ist, `u` ist die Eingangsgröße, `y` die Ausgangsgröße.

Auch Mehrgrößensysteme sind berechenbar; sind `u` und `y` Listen mit Variablen, so wird die entsprechende Übertragungsmatrix berechnet.



$$\begin{aligned} x_1 &= F_1 \cdot u \\ x_2 &= F_2 \cdot (u - x_3) \\ x_3 &= F_3 \cdot y \\ y &= x_1 + x_2 \end{aligned}$$

Aufstellen eines Gleichungssystems aus dem Blockschaltbild

$$\begin{aligned} (\%i14) \quad & \text{eqs} : [x1=F1*u, x2=F2*(u-x3), x3=F3*y, y=x1+x2]; \\ (\%o14) \quad & [x1=F1 u, x2=F2 (u-x3), x3=F3 y, y=x2+x1] \end{aligned}$$

Berechnung der Übertragungsfunktion aus dem Gleichungssystem

$$\begin{aligned} (\%i15) \quad & \text{transfer_function}(eqs,[x1,x2,x3,y],u,y); \\ (\%o15) \quad & \frac{F2 + F1}{F2 F3 + 1} \end{aligned}$$

`sum_form(F(s),n)` dividiert den Zähler und Nenner von $F(s)$ in Abhängigkeit von n durch einen bestimmten Koeffizienten, sodass dieser zu 1 wird:

- $n = 1$... führender Zählerkoeffizient von $F(s)$
- $n = 2$... niedrigster Zählerkoeffizient von $F(s)$
- $n = 3$... führender Nennerkoeffizient von $F(s)$
- $n = 4$... niedrigster Nennerkoeffizient von $F(s)$ (default)

```
(%i16) F: (2*s+3)/(4*s^3+5*s^2+6*s+7);
```

```
(%o16) 
$$\frac{2s+3}{4s^3+5s^2+6s+7}$$

```

Standardformen, die den führenden oder niedrigsten Zählerkoeffizienten zu 1 machen:

```
(%i17) [sum_form(F,1),sum_form(F,2)];
```

```
(%o17) 
$$\left[ \frac{s+1.5}{2s^3+2.5s^2+3s+3.5}, \frac{0.66667s+1}{1.3333s^3+1.6667s^2+2s+2.3333} \right]$$

```

Standardformen, die den führenden oder niedrigsten Nennerkoeffizienten zu 1 machen:

```
(%i18) [sum_form(F,3),sum_form(F,4)];
```

```
(%o18) 
$$\left[ \frac{0.5s+0.75}{s^3+1.25s^2+1.5s+1.75}, \frac{0.28571s+0.42857}{0.57143s^3+0.71429s^2+0.85714s+1} \right]$$

```

`product_form(F(s),n)` stellt die Übertragungsfunktion in faktorisierter Form dar: Zähler und Nenner werden in lineare und quadratische Faktoren aufgespalten:

Produktdarstellung einer Übertragungsfunktion

```
(%i19) product_form(F);
```

```
(%o19) 
$$\frac{0.42857 (0.66667s+1.0)}{(0.82799s+1.0) (0.69014s^2+0.029158s+1.0)}$$

```

4 Laplace Transformation, Sprungantwort

Zur Laplace-Transformation stellt Maxima die Funktion `laplace(f,t,s)` zur Verfügung, zur Rücktransformation `ilt(f,s,t)`. Die Koeffizienten der Zähler- und Nennerpolynome können dabei auch symbolische Werte haben. Allerdings versagt `ilt` bereits bei Nennerpolynomen 3. Grades, wenn keine Nullstellen analytisch gefunden werden.

Bei der in COMA enthaltenen Funktion `nilt` werden die Nullstellen des Nennerpolynoms mit `allroots` numerisch berechnet, es können daher gebrochen rationale Funktionen (fast) beliebiger Ordnung rücktransformiert werden, die Polynomkoeffizienten müssen dann jedoch reine Zahlenwerte sein.

<code>laplace(ft, timevar, lapvar)</code>	Laplace-Transformation der Funktion <i>ft</i> (Bestandteil von Maxima)
<code>ilt(fs, lapvar, timevar)</code>	Laplace-Rücktransformation von <i>fs</i> (Bestandteil von Maxima)
<code>nilt(fs, lapvar, timevar)</code>	Laplace-Rücktransformation von <i>fs</i> mit numerisch berechneten Polstellen
<code>step_response(F(s), opts)</code>	Plotten der Sprungantwort der Übertragungsfunktion <i>F(s)</i>

Laplace-Transformation

Laplace-Transformierte einer Funktion

(%i1) `laplace(t^2*sin(a*t),t,s);`

(%o1)
$$\frac{8 a s^2}{(s^2 + a^2)^3} - \frac{2 a}{(s^2 + a^2)^2}$$

Laplace-Rücktransformation

(%i2) `ilt(1/(s^3+2*s^2+2*s+1),s,t);`

(%o2)
$$e^{-\frac{t}{2}} \left(\frac{\sin\left(\frac{\sqrt{3}t}{2}\right)}{\sqrt{3}} - \cos\left(\frac{\sqrt{3}t}{2}\right) \right) + e^{-t}$$

Die Koeffizienten können auch
symbolisch sein

(%i3) `ilt(1/((s+a)^2*(s+b)),s,t);`

(%o3)
$$\frac{e^{-b t}}{b^2 - 2 a b + a^2} + \frac{t e^{-a t}}{b - a} - \frac{e^{-a t}}{b^2 - 2 a b + a^2}$$

Die Rücktransformation versagt, wenn
keine Nullstellen des Nenners analytisch
gefunden werden können.

(%i4) `ilt(1/(s^3+2*s^2+3*s+1),s,t);`

(%o4) `ilt(1/(s^3+2*s^2+3*s+1),s,t)`

`nilt` ermittelt die Nenner-Nullstellen
numerisch, Nennerpolynome beliebiger
Ordnung sind daher möglich.


(%i5) `nilt(1/(s^3+2*s^2+3*s+1),s,t);`

(%o5)
$$-0.14795 e^{-0.78492 t} \sin(1.3071 t) - 0.54512 e^{-0.78492 t} \cos(1.3071 t) + 0.54512 e^{-0.43016 t}$$

Aufbau einer Liste aus PT2-Elementen mit steigendem Dämpfungsgrad

```
(%i6) pt2li:create_list(1/(s^2+2*d*s+1), d,
[0.0001,0.1,0.2,0.3,0.4,0.5]);
(%o6) [1/(s^2+2.0 10^-4 s+1), 1/(s^2+0.2 s+1), 1/(s^2+0.4 s+1),
1/(s^2+0.6 s+1), 1/(s^2+0.8 s+1), 1/(s^2+1.0 s+1)]
```

Plotten der Sprungantworten; wird die Option xrange nicht angegeben, so erfolgt die Wahl des dargestellten Zeitbereichs automatisch.




PT1-Element mit zusätzlicher Totzeit in Padénäherung

```
(%i8) F:time_delay(2,5)*1/(1+s);
(%o8) (5 s^4 - 60 s^3 + 315 s^2 - 840 s + 945) / ((s + 1) (2 s^5 + 25 s^4 + 150 s^3 + 525 s^2 + 1050 s + 945))
```

Berechnung der exakten Sprungantwort des totzeitbehafteten PT1-Elements

```
(%i9) ft:unit_step(t-2)*ev(ilt(1/(s*(1+s)),s,t),
t=t-2);
(%o9) unit_step(t-2) (1-%e^(-t))
```

Vergleich der exakten Sprungantwort mit der Padénäherung; die berechnete exakte Sprungantwort ist als Grafikobjekt explicit angegeben.



5 Frequenzgänge

<code>bode_plot(F(s),opts)</code>	Bodediagramm von $F(j\omega)$
<code>magnitude_plot(F(s),opts)</code>	Amplitudengang des Bodediagramms von $F(j\omega)$
<code>logy=false</code>	Option für <code>magnitude_plot</code> , bewirkt eine <i>lineare</i> Skalierung des Amplitudengangs
<code>phase_plot(F(s),opts)</code>	Phasengang des Bodediagramms von $F(j\omega)$
<code>phase(F(s))</code>	Argument des Frequenzgangs $F(j\omega)$ in Grad
<code>asymptotic(F(s))</code>	Asymptotenknickzug des Frequenzgangs $F(j\omega)$
<code>nyquist_plot(F(s),opts)</code>	Frequenzgangsortskurve von $F(j\omega)$

Frequenzgänge

Als Parameter sind Übertragungsfunktionen $F(s)$ mit der Laplace-Variable s anzugeben, die von den Plotroutinen automatisch durch $j\omega$ ersetzt wird.

Werden in den Plotfunktionen die Optionen `xrange` und `yrange` nicht angegeben, so wird der dargestellte Bereich automatisch ermittelt. Mit den Optionen `logx=false` und `logy=false` können die Achsen von Bodediagrammen linear skaliert werden. `bode_plot` benötigt für die Option `yrange` eine *Liste mit zwei Bereichen*, jeweils einen für den Amplitudengang und einen für den Phasengang.

Bei der Ortskurve (`nyquist_plot`) ist die Skalierung in x-Richtung und in y-Richtung standardmäßig gleich (`aspect_ratio=-1`), was zu einer unverzerrten Darstellung führt.

Im Gegensatz zur Maxima-Funktion `carg`, die das Argument einer komplexen Zahl (im Bogenmaß) immer im Intervall $-\pi \dots \pi$ berechnet, ermittelt `phase` die tatsächliche Phasendrehung zwischen Eingangs- und Ausgangsgröße, die beliebig hohe Werte annehmen kann; jede Nullstelle und jede Polstelle bewirkt eine Phasendrehung um $\pi/2$ (bzw. 90 Grad) mit entsprechendem Vorzeichen.


Bei Auftreten von hohen Resonanzen findet in einem kleinen Frequenzbereich eine sehr große Phasendrehung statt. Um auch in solchen Fällen – insbesondere bei der Ortskurve – eine saubere glatte Kurve zu erhalten, muss unter Umständen die Anzahl der Stützstellen für die Plotroutine erhöht werden, was mit der Option `nticks=value` erfolgen kann (Defaultwert 500).

Liste aus PT2-Elementen mit steigendem Dämpfungsgrad

```
(%i1)  fli:create_list(1/(s^2+2*d*s+1), d,
                    [0.0001,0.1,0.2,0.3,0.4,0.5]);
(%o1)  [
          1
          s^2+2.0 10^-4 s+1, 1
          s^2+0.2 s+1, 1
          s^2+0.4 s+1,
          1
          s^2+0.6 s+1, 1
          s^2+0.8 s+1, 1
          s^2+1.0 s+1]
```


Bodediagramm der PT2-Elemente; die Bereiche für die y-Achsen sind in einer Liste anzugeben.

```
(%i2) bode_plot (fli, yrange=[[0.1,10],[-200,20]])$
```




Amplitudengänge der PT2-Elemente mit linearer Skalierung beider Achsen

```
(%i3) magnitude_plot (fli, xrange=[0,2],  
yrange=[0.1,5], logx=false, logy=false)$
```



Phasengänge der PT2-Elemente mit linearer Skalierung beider Achsen, die y-Achse ist hier standardmäßig linear skaliert.

```
(%i4) phase_plot (fli, xrange=[0,2], yrange=[-200,20],  
logx=false)$
```




Übertragungsfunktion eines PID-Reglers

(%i5) `Fr:2*(1+1/(2*s)+0.1*s/(1+0.02*s));`

(%o5)
$$2 \left(\frac{0.1 s}{0.02 s + 1} + \frac{1}{2 s} + 1 \right)$$


Bodediagramm des PID-Reglers; exakter Verlauf (rot) und Asymptotenknickzug (blau)

(%i6) `bode_plot ([Fr,asymptotic (Fr)],
xrange=[0.01,1000],yrange=[[1,100],[-120,120]])$`



Ortskurven der PT2-Elemente; es kann notwendig sein, die Anzahl der Stützstellen mit der Option `nticks` zu erhöhen.

(%i7) `nyquist_plot (fli,xrange=[-5,5],yrange=[-6,1],
nticks=2000)$`



Übertragungsfunktion dritter Ordnung.

(%i8) `f:2/(1+2*s+2*s^2+s^3);`

(%o8)
$$\frac{2}{s^3 + 2s^2 + 2s + 1}$$

Argument von $F(j\omega)$ durch Zerlegung des Frequenzgangs in lineare und quadratische Faktoren und Addition der Teilargumente.

```
(%i9) phase(f);
(%o9)  $\frac{180 \left( -\operatorname{atan}\left(1.0 \omega\right) - \operatorname{atan2}\left(1.0 \omega, 1.0 - 1.0 \omega^2\right) \right)}{\pi}$ 
```

Mit den Grafikobjekten `points` und `label` kann eine Ortskurve mit ω -Werten markiert und beschriftet werden; die Positionen der Beschriftungen können mit einigen trickreichen Überlegungen geeignet festgelegt werden: die Beschriftung eines Punktes liegt in einem bestimmten Abstand auf einer Normalen zum Richtungsvektor in diesem Punkt.

Beachte: Punkte und Vektoren sind hier als *Listen* (und nicht als *Matrizen*) definiert.

Liste mit ω -Werten zum Markieren der Ortskurve

```
(%i10) omegali: makelist(0.1*k, k, 1, 10);
(%o10) [0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 1.0]
```

Umwandlung in den Frequenzgang durch Ersetzen von s durch $j\omega$

```
(%i11) fom: ev(f, s=%i*omega);
(%o11)  $\frac{2}{-i \omega^3 - 2 \omega^2 + 2 i \omega + 1}$ 
```

Position eines Punktes auf der Ortskurve (Liste aus x- und y-Koordinate):

```
(%i12) dot: [realpart(fom), imagpart(fom)];
(%o12)  $\left[ \frac{2(1-2\omega^2)}{(2\omega-\omega^3)^2 + (1-2\omega^2)^2}, \frac{2(\omega^3-2\omega)}{(2\omega-\omega^3)^2 + (1-2\omega^2)^2} \right]$ 
```

Die Ableitung gibt einen Richtungsvektor auf der Ortskurve an.

```
(%i13) abl: ratsimp(diff(dot, omega));
(%o13)  $\left[ \frac{16\omega^7 - 12\omega^5 - 8\omega}{\omega^{12} + 2\omega^6 + 1}, -\frac{6\omega^8 - 20\omega^6 - 6\omega^2 + 4}{\omega^{12} + 2\omega^6 + 1} \right]$ 
```

Einheitsvektor normal auf die Ortskurve

```
(%i14) ovec: ratsimp([-abl[2], abl[1]]/
sqrt(abl[1]^2+abl[2]^2));
(%o14)  $\left[ \frac{6\omega^8 - 20\omega^6 - 6\omega^2 + 4}{\sqrt{36\omega^4 + 16\omega^2 + 16} \sqrt{\omega^{12} + 2\omega^6 + 1}}, \frac{16\omega^7 - 12\omega^5 - 8\omega}{\sqrt{36\omega^4 + 16\omega^2 + 16} \sqrt{\omega^{12} + 2\omega^6 + 1}} \right]$ 
```

Position der Beschriftung für einen Punkt

```
(%i15) lab: dot + 0.3*ovec$
```

Die zu markierenden Ortskurvenpunkte werden als Grafikobjekt `points` definiert.

```
(%i16) punkte: points(map(lambda([u], ev(dot, omega=u)),
omegali))$
```

Die zugehörigen Beschriftungen werden als Grafikobjekt `label` definiert.

```
(%i17) labs: apply(label, map(lambda([u], [string(u),
ev(lab[1], omega=u), ev(lab[2], omega=u)]),
omegali))$
```


Einheitskreis als Kurve in Parameterdarstellung

```
(%i18) circle: parametric(cos(t), sin(t), t, 0, 2*pi);
(%o18) parametric(cos(t), sin(t), t, 0, 2*pi)
```

Ortskurve mit markierten
Ortskurvenpunkten und dem
Einheitskreis

```
(%i19) nyquist_plot ([circle,f,punkte,labels],  
                    xrange=[-2,3],yrange=[-2.5,0.5],  
                    point_type=7,color=[black,red,red,black])$
```

```
(%t19)
```



6 Untersuchungen in der s-Ebene

6.1 Pol/Nullstellen-Verteilung

<code>poles(F(s))</code>	Polstellen der Übertragungsfunktion $F(s)$
<code>zeros(F(s))</code>	Nullstellen der Übertragungsfunktion $F(s)$
<code>poles_and_zeros(F(s),opts)</code>	Darstellung der Pol/Nullstellen-Verteilung einer Übertragungsfunktion $F(s)$ in der s-Ebene

Pol/Nullstellen-Verteilung

Die Funktionen `poles` und `zeros` geben die Pol- bzw. Nullstellen der Übertragungsfunktion als Liste aus, `poles_and_zeros` zeichnet die Pol- und Nullstellen in der s-Ebene. Dabei werden die Polstellen durch ein \times , die Nullstellen durch ein \circ dargestellt. Standardmäßig ist die Skalierung in x-Richtung und in y-Richtung gleich (`aspect_ratio=-1`), was zu einer unverzerrten Darstellung führt.

Erzeugung einer Liste aus
Zufalls-Übertragungsfunktionen

```
(%i1) fli: makelist (stable_rantranf (5), k, 1, 2);
```

```
(%o1) [ 
$$\frac{7s^4 + 2s^3 + 4s^2 + 8s + 4}{2s^5 + 4s^4 + 7s^3 + 5s^2 + 5s + 1},$$
 
$$\frac{8s + 1}{2s^5 + 4s^4 + 8s^3 + 10s^2 + 3s + 2}$$
 ]
```

Nullstellen

```
(%i2) zeros (fli);
```

```
(%o2) [ [0.2688 %i-0.61289, -0.2688 %i-0.61289, 0.47003-1.0271 %i, 1.0271 %i+0.47003], [-0.125] ]
```


Polstellen

```
(%i3) poles (fli);
```

```
(%o3) [ [-0.2408, 1.0 %i, -1.0 %i, -1.1414 %i-0.8796, 1.1414 %i-0.8796], [0.49432 %i-0.079394, -0.49432 %i-0.079394, -1.6717 %i-0.21887, -1.4035, 1.6717 %i-0.21887] ]
```

Pol/Nullstellenverteilungen in der komplexen s-Ebene

```
(%i4) poles_and_zeros (fli)$
(%t4)
```



6.2 Wurzelortskurven

<code>root_locus(F(s,k),opts)</code>	Wurzelortskurve einer Übertragungsfunktion $F(s, k)$ mit einem freien Parameter k in der s-Ebene
<code>trange=[min,max]</code>	Bereich für den freien Parameter k , default: $[0.001, 100]$
<code>nticks=n</code>	Anzahl der berechneten Ortskurvenpunkte, default: 500

Wurzelortskurven

`root_locus` zeichnet die Wurzelortskurve einer Übertragungsfunktion $F(s)$ in Abhängigkeit eines freien Parameters k , der aber nicht (wie bei den „klassischen“ Wurzelortskurven) die Verstärkung der offenen Regelschleife sein muss, sondern ein beliebiger Parameter, der die Übertragungsfunktion beeinflusst. Werden mehrere Übertragungsfunktionen angegeben, können die Namen der Parameter unterschiedlich sein, der Parameterbereich ist aber für alle Übertragungsfunktionen gleich, angegeben durch die Option `trange`.

Die Berechnung der Ortskurvenpunkte erfolgt über den Parameterbereich logarithmisch verteilt, der Parameter darf daher nur positive Werte annehmen.

Die Anfangspunkte der Wurzelortskurven werden durch ein \times , die Endpunkte durch ein \circ dargestellt. Wenn der Parameter die Verstärkung der offenen Regelschleife $F_O(s)$ ist, sein Startwert hinreichend klein und sein Endwert hinreichend groß gewählt wird, entspricht das den Polstellen bzw. den Nullstellen von $F_O(s)$.


Erzeugung einer Liste von Regelkreisen aus Übertragungsfunktionen mit jeweils einer anderen Nullstelle a und der veränderlichen Verstärkung k

```
(%i5) fli:closed_loop(makelist(k*((s-a)*(s+1))
/(s*(s-2)*(s+7)),a,-11,-8));
```

```
(%o5) [  $\frac{k s^2 + 12 k s + 11 k}{s^3 + k s^2 + 5 s^2 + 12 k s - 14 s + 11 k}$ ,
 $\frac{k s^2 + 11 k s + 10 k}{s^3 + k s^2 + 5 s^2 + 11 k s - 14 s + 10 k}$ ,
 $\frac{k s^2 + 10 k s + 9 k}{s^3 + k s^2 + 5 s^2 + 10 k s - 14 s + 9 k}$ ,
 $\frac{k s^2 + 9 k s + 8 k}{s^3 + k s^2 + 5 s^2 + 9 k s - 14 s + 8 k}$  ]
```


Wurzelortskurven in Abhängigkeit der „wandernden“ Nullstelle

```
(%i6) root_locus(fli,xrange=[-17,3],
yrange=[-6,6],nticks=5000)$
```



Wurzelortskurve eines PT2-Elements mit dem Dämpfungsgrad als Parameter

```
(%i7) root_locus(1/(s**2+a*s+1),xrange=[-4,1],
trange=[1e-4,3],nticks=5000)$
```



7 Stabilitätsverhalten

7.1 Stabilität

<code>stablep(F(s))</code>	Überprüft das System mit der Übertragungsfunktion $F(s)$ auf Stabilität
<code>stability_limit(F(s), k)</code>	Ermittlung der Stabilitätsgrenze der Übertragungsfunktion $F(s, k)$ in Bezug auf den Parameter k
<code>hurwitz(p(s))</code>	Berechnung der Hurwitz-Determinanten des Polynoms $p(s)$
<code>stable_area(F(s, a, b), a, b, opts)</code>	Darstellung des Stabilitätsbereichs der Übertragungsfunktion $F(s, a, b)$ in der a/b -Parameterebene

Stabilität

Die Funktion `stability_limit(F(s), k)` liefert Bedingungen in der Form

$$k = \text{value}, \omega = \text{value},$$

für rein imaginäre Pole, was für gängige Systeme mit der Stabilitätsgrenze gleichzusetzen ist. Dabei ist der Wert von ω die Kreisfrequenz der Dauerschwingung an der Stabilitätsgrenze. Diese Bedingungen können auch für *mehrere* Werte von ω erfüllt sein. Welche Werte tatsächlich Grenzstabilität bedeuten, ist durch zusätzliche Überlegungen zu ermitteln.

Eine exakte Stabilitätsaussage liefert das *Hurwitz-Kriterium*: Alle Nullstellen des Polynoms $p(s)$ haben genau dann einen negativen Realteil (d. h. die Übertragungsfunktion mit dem Nenner $p(s)$ ist genau dann stabil), wenn alle Hurwitz-Determinanten einen Wert größer Null ergeben. Die Funktion `hurwitz(p(s))` liefert eine Liste der Hurwitz-Determinanten, die Koeffizienten von $p(s)$ können dabei auch symbolische Werte haben.

`stable_area` stellt die Stabilitätsgrenze einer Übertragungsfunktion in Abhängigkeit zweier Parameter a und b in der a/b -Parameterebene dar. Werden die Optionen `xrange` und `yrange` nicht angegeben, so erfolgt die Darstellung jeweils im Bereich $0 \dots 1$.

Zufalls-Übertragungsfunktion

```
(%i1) f:stable_rantrnf(5);
```

$$(\%o1) \quad \frac{9}{s^5 + 3s^4 + 9s^3 + 10s^2 + 7s + 6}$$

Berechnung des geschlossenen Regelkreises mit einer Reglerverstärkung k

```
(%i2) fw:closed_loop(k*f);
```

$$(\%o2) \quad \frac{9k}{s^5 + 3s^4 + 9s^3 + 10s^2 + 7s + 9k + 6}$$

Ermittlung der Stabilitätsgrenze;
mehrere Bedingungen für imaginäre
Polstellen sind möglich.

```
(%i3)  lim:stability_limit (fw,k),numer;
(%o3)  [[k=-13.709 ,ω=2.8531] , [k=0.042326 ,ω=
0.92733 ]]
```

Das Hurwitz-Kriterium liefert exakte
Aussagen; der Regelkreis ist genau dann
stabil, wenn alle Listenelemente positiv
sind.

```
(%i4)  ratsimp (hurwitz (denom (fw)));
(%o4)  [16-81 k, 39-666 k, -81 k2-1107 k+47, -81 k2-
1107 k+47]
```

Erstellung einer Liste von
Verstärkungen: oberhalb, an und
unterhalb der Stabilitätsgrenze

```
(%i5)  kli:float (ev ([1.1*k,k,0.9*k],second (lim)));
(%o5)  [0.046559 , 0.042326 , 0.038093 ]
```

Berechnung der Übertragungsfunktionen
der zugehörigen offenen Regelschleifen,

```
(%i6)  foli:create_list (k*f,k,kli);
(%o6)  [

$$\frac{0.41903}{s^5 + 3s^4 + 9s^3 + 10s^2 + 7s + 6},$$


$$\frac{0.38093}{s^5 + 3s^4 + 9s^3 + 10s^2 + 7s + 6}, \frac{0.34284}{s^5 + 3s^4 + 9s^3 + 10s^2 + 7s + 6}]$$

```

sowie der geschlossenen Regelkreise

```
(%i7)  fwli:(closed_loop (foli))$
```

Stabilitätsprüfung, stablep liefert hier
für den grenzstabilen Fall true

```
(%i8)  stablep (fwli);
(%o8)  [false , true , true]
```

Kontrolle der Rechnung an Hand der
Sprungantworten, die Periodendauer der
Dauerschwingung an der
Stabilitätsgrenze gemäß $T = 2\pi/\omega$
ergibt ungefähr den Wert 6.7.



Liste von PT5-Elementen mit zwei freien
Parametern a und b

```
(%i10)  fli:makelist (1/(s5+3*s4+k*s3+a*s2+b*s+1)
,k,2,5);
```

```
(%o10)  [


$$\frac{1}{s^5 + 3s^4 + 2s^3 + as^2 + bs + 1},$$


$$\frac{1}{s^5 + 3s^4 + 3s^3 + as^2 + bs + 1}, \frac{1}{s^5 + 3s^4 + 4s^3 + as^2 + bs + 1},$$


$$\frac{1}{s^5 + 3s^4 + 5s^3 + as^2 + bs + 1}]$$

```


Darstellung des Stabilitätsbereichs in Abhängigkeit der Parameter a und b



7.2 Stabilitätsgüte

Die Beurteilung der Stabilitätsgüte kann mit Hilfe des Phasenrandes α_R oder des Amplitudenrandes A_r erfolgen. Die zugehörigen Kreisfrequenzen sind die Durchtrittsfrequenz ω_D bzw. die Phasenschnittfrequenz ω_r .

<code>gain_crossover(F(s))</code>	Berechnung der Durchtrittsfrequenzen ω_D von $F(j\omega)$ für die gilt: $ F(j\omega_D) = 1$
<code>phase_margin(F(s))</code>	Berechnung des Phasenrandes α_R von $F(j\omega)$ in Grad
<code>phase_crossover(F(s))</code>	Berechnung der Phasenschnittfrequenz ω_r von $F(j\omega)$ für die gilt: $\arg F(j\omega_r) = -\pi$
<code>gain_margin(F(s))</code>	Berechnung des Amplitudenrandes A_R von $F(j\omega)$
<code>damping(F(s))</code>	Dämpfung von $F(j\omega)$ (negativer Realteil des rechtsten Pols)
<code>damping_ratio(F(s))</code>	minimaler Dämpfungsgrad aller Polpaare von $F(j\omega)$

Stabilitätsgüte


`gain_crossover` liefert eine Liste von Durchtrittsfrequenzen ω_D , bei denen der Betrag des Frequenzgangs gleich 1 ist. `phase_margin` liefert die zugehörigen Phasenränder, die Differenzen auf -180° . Welcher davon der für die Stabilität maßgebliche Phasenrand ist, ist im Einzelfall zu beurteilen (es könnten prinzipiell auch mehrere sein).

Durchtrittsfrequenzen ω_D , können (insbesondere bei Resonanz) auch mehrfach auftreten.

```
(%i12) gain_crossover (f01i);
(%o12) [[ $\omega=0.89261$ ,  $\omega=0.93126$ ], [ $\omega=0.89673$ ,  $\omega=$ 
 $0.92733$ ], [ $\omega=0.90173$ ,  $\omega=0.92252$ ]]
```

Amplitudengänge mit mehreren Durchtrittsfrequenzen

```
(%i13) magnitude_plot (fcli,xrange=[0.8,1.1],
                       yrange=[0,1.5],logx=false,logy=false)$
(%t13)
```



Phasenränder α_R in Grad, instabile Regelkreise haben einen negativen Wert.

```
(%i14) phase_margin (fcli);
(%o14) [[81.347, -6.7748], [74.56, -5.1512 10-5], [64.484, 10.065]]
```

Phasenschnittfrequenzen ω_r , können auch mehrfach auftreten.

```
(%i15) phase_crossover (fcli);
(%o15) [[ $\omega = \frac{\sqrt{9-\sqrt{53}}}{\sqrt{2}}$ ], [ $\omega = \frac{\sqrt{9-\sqrt{53}}}{\sqrt{2}}$ ], [ $\omega = \frac{\sqrt{9-\sqrt{53}}}{\sqrt{2}}$ ]]
```

Amplitudenränder A_R , instabile Regelkreise haben einen Wert kleiner 1.

```
(%i16) gain_margin (fcli);
(%o16) [[0.90909], [1.0], [1.1111]]
```

damping berechnet die *Dämpfung* σ einer Übertragungsfunktion. Das ist der negative Realteil des am weitesten rechts liegenden Pols oder Polpaars. Sie gibt an, mit welcher Geschwindigkeit ein transienter Vorgang abklingt (oder aufklingt). Der *Dämpfungsgrad* D ist die relative Dämpfung, bezogen auf die natürliche Kreisfrequenz ω_n eines Polpaars. `damping_ratio` berechnet den minimalen Dämpfungsgrad aller Polpaare einer Übertragungsfunktion. σ und D haben positive Werte bei stabilen Übertragungsfunktionen und negative Werte bei instabilen Übertragungsfunktionen

Dämpfung dreier Übertragungsfunktionen

```
(%i17) damping (fwli);
(%o17) [-0.0020036, 0, 0.0020153]
```

Dämpfungsgrad dreier Übertragungsfunktionen

```
(%i18) damping_ratio (fwli);
(%o18) [-0.0021573, 0.0, 0.0021766]
```

8 Optimierung

Das Güteintegral nach dem ISE-Kriterium kann gemäß dem Parseval-Theorem direkt im Laplace-Bereich berechnet werden:

$$I_{ISE} = \int_0^{\infty} e^2(t) dt = \frac{1}{2\pi j} \int_{-j\infty}^{j\infty} E(s) \cdot E(-s) ds$$

Dabei ist $e(t)$ eine mit steigender Zeit gegen Null strebende Funktion, bei Optimierungsaufgaben üblicherweise die Regelabweichung. Gemäß [6] gelingt die Berechnung des Integrals rein algebraisch.

`ise(E(s))` Güteintegral der Funktion $e(t)$ nach dem ISE-Kriterium

Integrales Gütemaß

Ableiten des Integrals nach freien Parametern (z. B. Reglerparametern) und Nullsetzen der Ableitungen liefert die optimalen Werte der Parameter.

Übertragungsfunktion mit zwei freien Parametern a und b

```
(%i1) f:1/(s**3+a*s**2+b*s+1);
(%o1) 
$$\frac{1}{s^3 + a s^2 + b s + 1}$$

```

Berechnung der Abweichung vom Stationärwert bei Anregung mit der Sprungfunktion

```
(%i2) xs:ratsimp((1-f)/s);
(%o2) 
$$\frac{s^2 + a s + b}{s^3 + a s^2 + b s + 1}$$

```

Güteintegral nach dem ISE-Kriterium

```
(%i3) iise:ise(xs);
(%o3) 
$$\frac{a b^2 - b + a^2}{2 a b - 2}$$

```

Ableiten nach den Parametern a und b (Berechnung der Jacobimatrix)

```
(%i4) abl:ratsimp(jacobian([iise],[a,b]));
(%o4) 
$$\begin{bmatrix} a^2 b - 2 a & a^2 b^2 - 2 a b - a^3 + 1 \\ 2 a^2 b^2 - 4 a b + 2 & 2 a^2 b^2 - 4 a b + 2 \end{bmatrix}$$

```

Beschränkung auf reelle Lösungen beim Lösen von Gleichungssystemen

```
(%i5) realonly:true;
(%o5) true
```

Auflösen nach a und b , die Ableitungen werden hier automatisch 0 gesetzt.

```
(%i6) res:solve(abl[1],[a,b]);
(%o6) [[a=1,b=2]]
```

Einsetzen in f liefert die „optimale“ Übertragungsfunktion.

```
(%i7) fopt:ev(f,res);
(%o7) 
$$\frac{1}{s^3 + s^2 + 2 s + 1}$$

```

9 Reglerentwurf

`gain_optimum(Fs(s),Fr(s))` Dimensionierung eines Reglers nach dem Betragsoptimum.

Reglerentwurfsverfahren

`gain_optimum` ermittelt einen betragsoptimalen Regler $F_R(s)$ zu einer gegebenen Regelstrecke $F_S(s)$. Die Struktur des Reglers und die Namen für die Reglerparameter sind grundsätzlich frei wählbar. Ob tatsächlich Lösungen für die Reglerparameter gefunden werden, hängt natürlich von der Sinnhaftigkeit der Wahl ab (ein PT1-Regler wird beispielsweise vermutlich keine Lösung ergeben).

Übertragungsfunktion einer Regelstrecke

```
(%i1) fs:2/((1+5*s)*(1+s)**2*(1+0.3*s));
```

$$(\%o1) \frac{2}{(0.3s+1)(s+1)^2(5s+1)}$$

Liste bestehend aus einem I-, PI- und einem PID-Regler

```
(%i2) [fri,frpi,frpid]:[1/(s*Ti),
kr*(1+1/(s*Tn)),(1+s*Ta)*(1+s*Tb)/(s*Tc)];
```

$$(\%o2) \left[\frac{1}{Ti s}, kr \left(\frac{1}{Tn s} + 1 \right), \frac{(Ta s + 1)(Tb s + 1)}{Tc s} \right]$$

Betragsoptimum für den I-Regler

```
(%i3) g1:gain_optimum(fs,fri);
```

$$(\%o3) [Ti = \frac{146}{5}, Ti = 0]$$

Betragsoptimum für den PI-Regler; die Reglernullstelle kompensiert näherungsweise die größte Streckenzeitkonstante.

```
(%i4) g2:gain_optimum(fs,frpi);
```

$$(\%o4) [kr = \frac{206057}{349320}, Tn = \frac{206057}{40190}]$$

Betragsoptimum für den PID-Regler; die Reglernullstellen kompensieren näherungsweise die beiden größten Streckenzeitkonstanten.

```
(%i5) g3:float(gain_optimum(fs,frpid));
```

$$(\%o5) [Tc = 3.6197, Ta = 4.9984, Tb = 1.3966]$$


Einsetzen der Ergebnisse in die Regler-Übertragungsfunktionen

```
(%i6) reli:float(ev([fri,frpi,frpid],[g1,g2,g3]));
```

$$(\%o6) \frac{\left[\frac{0.034247}{s}, 0.58988 \left(\frac{0.19504}{s} + 1.0 \right), 0.27627 \frac{(1.3966 s + 1.0)(4.9984 s + 1.0)}{s} \right]}{s}$$

Die Sprungantworten bestätigen ca. 5% Überschwingen und Anregelzeiten im Ausmaß von etwa 4,7 mal der Summe der verbleibenden Zeitkonstanten.

```
(%i7) step_response (float(ev(closed_loop (reli*fs),
                                res)),yrange=[0,1.5])$
(%t7)
```



Die Regelstrecke kann auch symbolische Koeffizienten haben.

```
(%i8) fs:2/((1+a*s)*(1+s**2)*(1+b*s));
```

```
(%o8) 
$$\frac{2}{(a s + 1) (b s + 1) (s^2 + 1)}$$

```

Das Ergebnis sind Formeln für die optimalen Reglerparameter.

```
(%i9) gain_optimum (fs,frpi);
```

```
(%o9) [kr= $\frac{b^2+a^2-1}{4ab}$ , Tn= $\frac{b^3+ab^2+(a^2-1)b+a^3-a}{b^2+ab+a^2-1}$ ]
```


10 Zustandsraum

<code>System: [A,B,C,D]</code>	Definition eines linearen Systems als Liste aus den Zustandsmatrizen A , B , C und D
<code>systemp(A,B,C[,D])</code> <code>systemp(System)</code>	Überprüft, ob <i>System</i> ein gültiges lineares System aus Zustandsmatrizen ist.
<code>nsystemp(A,B,C[,D])</code> <code>nsystemp(System)</code>	Überprüft, ob <i>System</i> ein gültiges lineares System ist, bei dem alle Matrizenelemente Zahlenwerte sind.
<code>transfer_function(A,B,C[,D])</code> <code>transfer_function(System)</code>	Berechnung der Übertragungsfunktion (Übertragungsmatrix) aus der Zustandsraumdarstellung
<code>controller_canonical_form(f)</code>	Berechnung der Regelungsnormalform aus der Übertragungsfunktion f
<code>observer_canonical_form(f)</code>	Berechnung der Beobachtungsnormalform aus der Übertragungsfunktion f
<code>controllability_matrix(A,B)</code> <code>controllability_matrix(System)</code>	Berechnung der Steuerbarkeitsmatrix
<code>observability_matrix(A,C)</code> <code>observability_matrix(System)</code>	Berechnung der Beobachtbarkeitsmatrix

Zustandsraumdarstellung

Unter *System* ist eine Zusammenfassung der vier Zustandsmatrizen **A** (Systemmatrix), **B** (Eingangsmatrix), **C** (Ausgangsmatrix) und **D** (Durchgangsmatrix) in einer Liste zu verstehen. Bei Systemen ohne Durchgriff kann **D** entfallen, bei Eingrößensystemen kann **D** ein Skalar d sein. Allen Funktionen, denen ein *System* als Parameter übergeben wird, können auch direkt die entsprechenden Zustandsmatrizen (ohne Zusammenfassung zu einer Liste) übergeben werden.

Elektrischer Vierpol mit Zustandsgleichungen:



$$\begin{aligned}\frac{di_1}{dt} &= \frac{1}{L} \cdot (u_e - R \cdot i_1 - u_{C1}) \\ \frac{di_2}{dt} &= \frac{1}{L} \cdot (u_{C1} - u_{C2} - R \cdot i_2) \\ \frac{du_{C1}}{dt} &= \frac{1}{C} \cdot (i_2 - i_1) \\ \frac{du_{C2}}{dt} &= \frac{1}{C} \cdot i_2\end{aligned}$$

Aus den Zustandsgleichungen ergeben sich direkt die Zustandsmatrizen **A**, **B** und **C**; ein direkter Durchgriff von der Eingangsspannung u_e zur Ausgangsspannung u_a ist nicht vorhanden, daher gilt $D = 0$, das daher weggelassen werden kann.

Systemmatrix **A** des Schaltkreises

```
(%i1) A:matrix([-R/L,0,-1/L,0],[0,-R/L,1/L,-1/L],
               [1/C1,-1/C1,0,0],[0,1/C1,0,0]);
```

```
(%o1)
```

$$\begin{bmatrix} -\frac{R}{L} & 0 & -\frac{1}{L} & 0 \\ 0 & -\frac{R}{L} & \frac{1}{L} & -\frac{1}{L} \\ \frac{1}{C1} & -\frac{1}{C1} & 0 & 0 \\ 0 & \frac{1}{C1} & 0 & 0 \end{bmatrix}$$

Eingangsmatrix **B** des Schaltkreises

```
(%i2) B:matrix([1/L],[0],[0],[0]);
```

```
(%o2)
```

$$\begin{bmatrix} \frac{1}{L} \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Ausgangsmatrix **C** des Schaltkreises

```
(%i3) C:matrix([0,0,0,1]);
```

```
(%o3)
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix}$$

Zusammenfassen der Zustandsmatrizen zu einem *System*

```
(%i4) circuit:[A,B,C];
```

```
(%o4)
```

$$\begin{bmatrix} -\frac{R}{L} & 0 & -\frac{1}{L} & 0 \\ 0 & -\frac{R}{L} & \frac{1}{L} & -\frac{1}{L} \\ \frac{1}{C1} & -\frac{1}{C1} & 0 & 0 \\ 0 & \frac{1}{C1} & 0 & 0 \end{bmatrix}, \begin{bmatrix} \frac{1}{L} \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix}$$

Die Liste mit den Zustandsmatrizen ist ein gültiges *System* ...

```
(%i5) systemp(circuit);
```

```
(%o5) true
```

... die Matrizen haben aber nicht ausschließlich Zahlenwerte als Elemente.

```
(%i6) nsystemp(circuit);
```

```
(%o6) false
```

`transfer_function` berechnet aus den Zustandsmatrizen die Übertragungsfunktion (oder die Übertragungsmatrix im Mehrgrößenfall). Diese Funktion ist in gewisser Weise polymorph: werden als Parameter keine Zustandsmatrizen sondern eine Liste von linearen Gleichungen und

Listen von Variablen übergeben, so wird aus diesen Gleichungen eine Übertragungsfunktion ermittelt (siehe Abschnitt 3).

Berechnung der Übertragungsfunktion;
sowohl die Angabe eines *Systems*, ...

```
(%i7) f:transfer_function (circuit);
(%o7)
```

$$\frac{1}{C_1^2 L^2 s^4 + 2 C_1^2 L R s^3 + C_1^2 R^2 s^2 + 3 C_1 L s^2 + 3 C_1 R s + 1}$$

... als auch die Angabe der einzelnen
Zustandsmatrizen ist möglich.

```
(%i8) transfer_function (A,B,C);
(%o8)
```

$$\frac{1}{C_1^2 L^2 s^4 + 2 C_1^2 L R s^3 + C_1^2 R^2 s^2 + 3 C_1 L s^2 + 3 C_1 R s + 1}$$

Die direkte Berechnung als
Impedanzkette liefert erwartungsgemäß
das selbe Ergebnis.

```
(%i9) f:impedance_chain (R+s*L,1/(s*C1),2);
(%o9)
```

$$\frac{1}{C_1^2 L^2 s^4 + 2 C_1^2 L R s^3 + (C_1^2 R^2 + 3 C_1 L) s^2 + 3 C_1 R s + 1}$$

Zustandsmatrizen können aus der Übertragungsfunktion gemäß der Regelungsnormalform oder der Beobachtungsnormalform berechnet werden:

Berechnung der Zustandsmatrizen nach
der Regelungsnormalform

```
(%i10) circ1:controller_canonical_form (f);
```

```
(%o10) [ [ 0 1 0 0 ] [ 0 ]
[ 0 0 1 0 ] [ 0 ]
[ 0 0 0 1 ] [ 0 ]
[ -1/C1^2 L^2 -3 R/C1 L^2 - (C1^2 R^2 + 3 C1 L)/C1^2 L^2 - 2 R/L ] [ 1 ] ] , [ 1/C1^2 L^2 0 0 0 ]
```

Berechnung der Zustandsmatrizen nach
der Beobachtungsnormalform

```
(%i11) circ2:observer_canonical_form (f);
```

```
(%o11) [ [ 0 0 0 -1/C1^2 L^2 ]
[ 1 0 0 -3 R/C1 L^2 ]
[ 0 1 0 - (C1^2 R^2 + 3 C1 L)/C1^2 L^2 ]
[ 0 0 1 -2 R/L ] ] , [ 1/C1^2 L^2 ] , [ 0 0 0 1 ] , 0]
```

Steuerbarkeitsmatrix

```
(%i12) h1:ratsimp(controllability_matrix (A,B));
```

```
(%o12) [ 1/L - R/L^2 C1 R^2 - L/C1 L^3 - C1 R^3 - 2 L R/C1 L^4 ]
[ 0 0 1/C1 L^2 - 2 R/C1 L^3 ]
[ 0 1/C1 L - R/C1 L^2 C1 R^2 - 2 L/C1^2 L^3 ]
[ 0 0 0 1/C1^2 L^2 ] ]
```


Beobachtbarkeitsmatrix

```
(%i13) h2:observability_matrix (circuit);
```

$$(\%o13) \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & \frac{1}{C_1} & 0 & 0 \\ 0 & -\frac{R}{C_1 L} & \frac{1}{C_1 L} & -\frac{1}{C_1 L} \\ \frac{1}{C_1^2 L} & \frac{R^2}{C_1 L^2} & -\frac{2}{C_1^2 L} & -\frac{R}{C_1 L^2} \end{bmatrix}$$

Das System ist steuerbar und beobachtbar.

```
(%i14) [rank(h1), rank(h2)];
```

```
(%o14) [4, 4]
```

11 Diverse Funktionen

Das Paket COMA stellt einige zusätzliche Funktionen bereit, die zwar nicht speziell regelungstechnischer Natur sind, aber den Anwender bei regelungstechnischen Berechnungen unterstützen können.

$R_1 \ // \ R_2$	Parallelschaltung zweier Widerstände R_1 und R_2
$Z \ /_ \varphi$	Polarkoordinatendarstellung einer komplexen Zahl $z = Z \angle \varphi$ (Z „cis“ φ)
$z \ cf$	Postfix-Operator zur Ausgabe einer komplexen Zahl z in Polarkoordinatendarstellung $Z \angle \varphi$
<code>chop(x)</code>	Ersetzen aller Zahlen im Ausdruck x , die kleiner als 10^{-10} sind, durch 0
<code>coefficient_list(p,x)</code>	Liste der Koeffizienten des Polynoms p in der Variablen x
<code>set_option(name=val,list)</code>	Setzen oder Hinzufügen eines Listenelements als Hasheintrag zur Liste $list$
<code>delete_option(name,list)</code>	Löschen des Hasheintrags mit dem Namen $name$ aus der Liste $list$
<code>option_exists(name,list)</code>	Überprüft, ob ein Hasheintrag mit dem Namen $name$ in der Liste $list$ existiert
<code>list_option_exists(name,list)</code>	Überprüft, ob der Hasheintrag mit dem Namen $name$ existiert und eine Liste ist

Diverse Funktionen

„//“ und „/_“ sind zwei in der Elektrotechnik gebräuchliche Operatoren zur Parallelschaltung elektrischer Widerstände bzw. zur Eingabe komplexer Größen in Polarkoordinaten (Versorform). Der Postfix-Operator `cf` gibt eine komplexe Größe in Versorform aus.

Parallelschaltung von Impedanzen	(%i1)	$R \ // \ 1/(s \cdot C) \ // \ s \cdot L + R1;$
	(%o1)	$\frac{L R s}{C L R s^2 + L s + R} + R1$
Versordarstellung von komplexen Größen	(%i2)	$[U1, U2, U3] : [230, 230/_240, 230/_120];$
	(%o2)	$[230, -199.19 \%i - 115.0, 199.19 \%i - 115.0]$
Ausgabe einer komplexen Größe in Versordarstellung	(%i3)	$U1 + U3 \ cf;$
	(%o3)	$230.0 \ /_ \ 60.0$

`coefficient_list` erstellt eine Liste von Polynomkoeffizienten in steigender Reihenfolge:

Polynom in der Variablen x

```
(%i4) p:5*(x+y)^2+a*x^5;
```

```
(%o4) 5 (y+x)^2 + a x^5
```

Liste mit den Polynomkoeffizienten

```
(%i5) coefficient_list (p,x);
```

```
(%o5) [5 y^2, 10 y, 5, 0, 0, a]
```

Die Funktion `chop` entfernt aus einem Ausdruck alle Zahlen, die kleiner als 10^{-10} sind. Damit lassen sich numerische Fehler „ausbügeln“:

Wenn die Numerik verrückt spielt, ...

```
(%i6) x3:expand((s^2-1.1*s+1.1)
```

```
        *(s^2+1.1*s+1.1)*(s^2+1));
```

```
(%o6) s^6+1.99 s^4+2.2204 10^-16 s^3+2.2 s^2+1.21
```

... können kleine (auf Grund von numerischen Fehlern) auftretende Zahlenwerte beseitigt werden.

```
(%i7) chop(x3);
```

```
(%o7) s^6+1.99 s^4+2.2 s^2+1.21
```

Ein assoziatives Array (Hash), bestehend aus Schlüssel-Wert-Paaren, wird in Maxima mit einer Liste realisiert. Es eignet sich unter anderem besonders zum Verwalten von Standardeinstellungen („Optionen“) und für benannte Parameter von Funktionen (z. B. für Grafikroutinen).

Einige Routinen erleichtern den Umgang mit assoziativen Arrays:

Liste mit Schlüssel-Wert-Paaren (Hash-Einträgen)

```
(%i8) opts:[color=blue,xrange=[0,10]];
```

```
(%o8) [color=blue,xrange=[0,10]]
```

Ändern eines Hash-Wertes

```
(%i9) set_option(color=red,opts);
```

```
(%o9) [xrange=[0,10],color=red]
```

Existiert ein Schlüssel nicht, so wird ein neuer Eintrag hinzugefügt.

```
(%i10) set_option(title="Test",opts);
```

```
(%o10) [xrange=[0,10],color=red,title=Test]
```

Entfernen eines Hash-Eintrags

```
(%i11) delete_option(color,opts);
```

```
(%o11) [xrange=[0,10],title=Test]
```

Überprüfen, ob ein Hash-Eintrag existiert

```
(%i12) option_exists(xrange,opts);
```

```
(%o12) true
```

Lesen eines Wertes aus dem Hash

```
(%i13) get_option(title,opts);
```

```
(%o13) Test
```

Zurückgabe eines Defaultwertes, wenn ein Schlüssel nicht existiert

```
(%i14) get_option(color,opts,red);
```

```
(%o14) red
```

Die Funktion `get_option` zum Lesen eines Wertes wäre prinzipiell nicht notwendig, da es zu diesem Zweck die Maxima-Funktion `assoc` gibt. Im Gegensatz zu `assoc` akzeptiert `get_option` auch Listen, die nicht nur Schlüssel-Wert-Paare enthalten, sondern auch beliebige Ausdrücke (wovon von anderen COMA-Funktionen intern Gebrauch gemacht wird).

Literaturverzeichnis

- [1] Maxima Development Team: *Maxima Manual Version 5.39.0*,
<http://maxima.sourceforge.net> 2016 .
- [2] Various Authors: *Gnuplot 4.6, An Interactive Plotting Program*,
<http://www.gnuplot.info> 2014 .
- [3] Rodríguez Riotorto, M.: *A Maxima-Gnuplot Interface*,
<http://riotorto.users.sourceforge.net/Maxima/gnuplot/index.html>.
- [4] Haager W.: *Computeralgebra mit Maxima*, Hanser München, Leipzig 2014.
- [5] Haager W.: *Regelungstechnik kompetenzorientiert*, Hölder Pichler Tempsky Wien 2016.
- [6] Newton G., Gould L., Kaiser J.: *Analytical Design of Linear Feedback Control*, Wiley, New York 1957.